

177 (combinations) Write a program to find the number of ways to partition $a+b$ things into a things in the left part and b things in the right part. Include recursive time.

After trying the question, scroll down to the solution.

§ The number of ways to partition $a+b$ things into a things and b things is $(a+b)! / (a! \times b!)$ where $!$ is the factorial function. First without time.

$$x := (a+b)! / (a! \times b!) \quad \Leftarrow$$

if $a=0$ **then** $x:=1$

else $a:=a-1$. $x := (a+b)! / (a! \times b!)$. $a:=a+1$. $x := x \times (a+b)/a$ **fi**

The assignment $x := (a+b)! / (a! \times b!)$ means $x' = (a+b)! / (a! \times b!)$ \wedge $a'=a$ \wedge $b'=b$. On the right side it is a recursive call. Stating it as an assignment makes the proof easy: just use the substitution law and simplify. The proof is by cases. First case:

$$\begin{aligned} & a=0 \wedge (x:=1) \Rightarrow (x := (a+b)! / (a! \times b!)) && \text{definition of assignment} \\ = & a=0 \wedge x'=1 \wedge a'=a \wedge b'=b \Rightarrow x' = (a+b)! / (a! \times b!) \wedge a'=a \wedge b'=b && \text{use } 0!=1 \\ = & \top \end{aligned}$$

Second case, starting with the right side:

$$\begin{aligned} & a \neq 0 \wedge (a:=a-1. x := (a+b)! / (a! \times b!). a:=a+1. x := x \times (a+b)/a) && \text{assignment} \\ = & a \neq 0 \wedge (a:=a-1. x := (a+b)! / (a! \times b!). a:=a+1. x' = x \times (a+b)/a \wedge a'=a \wedge b'=b) && \\ & \text{substitution law 3 times} \\ = & a \neq 0 \wedge x' = (a-1+b)! / ((a-1)! \times b!) \times (a+b)/a \wedge a'=a \wedge b'=b && \text{simplify} \\ = & a \neq 0 \wedge x' = (a+b)! / (a! \times b!) \wedge a'=a \wedge b'=b && \text{specialization} \\ \Rightarrow & x := (a+b)! / (a! \times b!) \end{aligned}$$

Now the time.

$$t' = t+a \quad \Leftarrow \quad \text{if } a=0 \text{ then } x:=1$$

else $a:=a-1$. $t:=t+1$. $t' = t+a$. $a:=a+1$. $x := x \times (a+b)/a$ **fi**

Proof by cases. First case:

$$\begin{aligned} & a=0 \wedge (x:=1) \Rightarrow t' = t+a && \text{definition of assignment} \\ = & a=0 \wedge x'=1 \wedge a'=a \wedge b'=b \wedge t'=t \Rightarrow t' = t+a \\ = & \top \end{aligned}$$

Second case, starting with the right side:

$$\begin{aligned} & a \neq 0 \wedge (a:=a-1. t:=t+1. t' = t+a. a:=a+1. x := x \times (a+b)/a) && \text{assignment} \\ = & a \neq 0 \wedge (a:=a-1. t:=t+1. t' = t+a. a:=a+1. x' = x \times (a+b)/a \wedge a'=a \wedge b'=b \wedge t'=t) && \\ & \text{substitution law 3 times} \\ = & a \neq 0 \wedge (t' = t+a. x' = x \times (a+1+b)/(a+1) \wedge a'=a+1 \wedge b'=b \wedge t'=t) && \text{sequential composition} \\ = & a \neq 0 \wedge (\exists x'', a'', b'', t'' \cdot t' = t+a \wedge x' = x'' \times (a''+1+b'') / (a''+1) && \\ & \wedge a' = a''+1 \wedge b' = b'' \wedge t' = t'') && \text{one point 4 times} \\ = & a \neq 0 \wedge t' = t+a && \text{specialization} \\ \Rightarrow & t' = t+a \end{aligned}$$

When refining $x := (a+b)! / (a! \times b!)$, there was no time variable. Adding the time variable, we cannot write this as an assignment, because that would mean $t'=t$. We can put the result and the timing together as

$$x' = (a+b)! / (a! \times b!) \wedge a'=a \wedge b'=b \wedge t'=t+a$$

or as

$$x := (a+b)! / (a! \times b!). t := t+a$$

Here is a solution that is symmetric in a and b .

$$x := (a+b)! / (a! \times b!) \quad \Leftarrow$$

if $a=0 \vee b=0$ **then** $x:=1$

else $a:=a-1$. $b:=b-1$. $x := (a+b)! / (a! \times b!)$.

$a:=a+1$. $b:=b+1$. $x := x/a/b \times (a+b-1) \times (a+b)$ **fi**

And its execution time is smaller: $a \downarrow b$.

Here is a solution with the same execution time and its recursion does not require a stack.

$$x' = (a+b)! / (a! \times b!) \wedge t' = t+a \downarrow b \quad \Leftarrow$$

$$x := 1. x' = x \times (a+b)! / (a! \times b!) \wedge t' = t+a \downarrow b$$

$$x' = x \times (a+b)! / (a! \times b!) \wedge t' = t + a \downarrow b \Leftarrow$$

if $a=0 \vee b=0$ **then** *ok*
else $x := x/a/b \times (a+b-1) \times (a+b)$. $a := a-1$. $b := b-1$. $t := t+1$.
 $x' = x \times (a+b)! / (a! \times b!) \wedge t' = t + a \downarrow b$ **fi**

Now, here is a **for**-loop solution. Define invariant

$$A k \equiv x = (a+k)! / (a! \times k!)$$

Then

$$x' = (a+b)! / (a! \times b!) \Leftarrow x := 1. A 0 \Rightarrow A' b$$

$$A 0 \Rightarrow A' b \Leftarrow \text{for } k := 0; ..b \text{ do } k: 0; ..b \wedge A k \Rightarrow A'(k+1) \text{ od}$$

$$k: 0; ..b \wedge A k \Rightarrow A'(k+1) \Leftarrow x := x \times (a+k+1) / (k+1)$$

with timing $t' = t+b$.

Finally, here are two functional solutions. Define

$$f = \langle a, b: \text{nat} \cdot (a+b)! / (a! \times b!) \rangle$$

Then

$$f a b = \text{if } a=0 \text{ then } 1 \text{ else } f(a-1) b \times (a+b) / a \text{ fi}$$

with execution time a . For execution time $a \downarrow b$

$$f a b = \text{if } a=0 \vee b=0 \text{ then } 1 \text{ else } f(a-1) (b-1) \times (a+b-1) \times (a+b) / a / b \text{ fi}$$